
Stages of a DSM-Based FEM Program

Introduction to FEM

Preprocessing:   problem definition

Processing:         problem solution

Postprocessing:  showing results

Problem Definition Data Structures

 

Introduction to FEM

Geometry Data Set:
         NodeCoordinates

Element Data Set:
         ElemTypes, ElemNodeLists,
         ElemMaterial, ElemFabrication

Degree of Freedom Data Set:
         FreedomTags, FreedomValues

Miscellaneous Processing Data Set:
         ProcessOptions

Example Structure to Illustrate Data Preparation
(one-element model)

Introduction to FEM

Example Structure to Illustrate Data Preparation
(Plate with Central Circular Hole)

Introduction to FEM

Example Structure to Illustrate Data Preparation
(Plate with Central Circular Hole)

Introduction to FEM

Mesh Plot Showing Element and Node Numbers

Introduction to FEM

Solution Processing Driver Module

Introduction to FEM

MembraneSolution[nodcoor_,eletyp_,elenod_,elemat_,
  elefab_,eleopt_,doftag_,dofval_]:= Module[{K,Kmod,u,f,sig,j,n,ns,
  supdof,supval,numnod=Length[nodcoor],numele=Length[elenod]},
  u=f=sig={};
  K=MembraneMasterStiffness[nodcoor,
          eletyp,elenod,elemat,elefab,eleopt];  K=N[K];
  ns=0; Do [Do [If[doftag[[n,j]]>0,ns++],{j,1,2}],{n,1,numnod}];
  supdof=supval=Table[0,{ns}];
  k=0; Do [Do [If[doftag[[n,j]]>0,k++;supdof[[k]]=2*(n-1)+j;
         supval[[k]]=dofval[[n,j]]],{j,1,2}],{n,1,numnod}];
  f=ModifiedNodeForces[supdof,supval,K,Flatten[dofval]];     
  Kmod=ModifiedMasterStiffness[supdof,K];
  u=Simplify[Inverse[Kmod].f]; u=Chop[u];
  f=Simplify[K.u]; f=Chop[f]; 
  sig=MembraneNodalStresses[nodcoor,
        eletyp,elenod,elemat,elefab,eleopt,u];
  sig=Chop[sig];
  Return[{u,f,sig}]; 
];

Master Stiffness Assembler Introduction to FEM

MembraneMasterStiffness[nodcoor_,eletyp_,elenod_,
  elemat_,elefab_,eleopt_]:=
  Module[{numele=Length[elenod],numnod=Length[nodcoor],numer,
   ne,eNL,eftab,neldof,i,n,Em,Ν,Emat,th,ncoor,Ke,K},
  K=Table[0,{2*numnod},{2*numnod}];  numer=eleopt[[1]];
  For [ne=1,ne<=numele,ne++,
       {type}=eletyp[[ne]]; 
       If [type!="Trig3"&&type!="Trig6"&&type!="Trig10"&& 
           type!="Quad4"&&type!="Quad9",  
           Print["Illegal element type,ne=",ne]; Return[Null]];
       eNL=elenod[[ne]]; n=Length[eNL];
       eftab=Flatten[Table[{2*eNL[[i]]-1,2*eNL[[i]]},{i,n}]];
       ncoor=Table[nodcoor[[eNL[[i]]]],{i,n}];           
       {Em,Ν}=elemat[[ne]];
       Emat=Em/(1-Ν^2)*{{1,Ν,0},{Ν,1,0},{0,0,(1-Ν)/2}};
       {th}=elefab[[ne]];
       If [type=="Trig3", Ke=Trig3IsoPMembraneStiffness[ncoor,
          {Emat,0,0},{th},{numer}] ]; 
       If [type=="Quad4", Ke=Quad4IsoPMembraneStiffness[ncoor,
          {Emat,0,0},{th},{numer,2}] ];
       If [type=="Trig6", Ke=Trig6IsoPMembraneStiffness[ncoor,
          {Emat,0,0},{th},{numer,3}] ];
       If [type=="Quad9", Ke=Quad9IsoPMembraneStiffness[ncoor,
          {Emat,0,0},{th},{numer,3}] ];
       If [type=="Trig10",Ke=Trig10IsoPMembraneStiffness[ncoor,
          {Emat,0,0},{th},{numer,3}] ];
       neldof=Length[Ke];
       For [i=1,i<=neldof,i++,ii=eftab[[i]];
         For [j=i,j<=neldof,j++,jj=eftab[[j]];
             K[[jj,ii]]=K[[ii,jj]]+=Ke[[i,j]] 
         ];
       ];
     ];
  Return[K]; 
];

Nodal Stress Recovery
Introduction to FEM

MembraneNodalStresses[nodcoor_,eletyp_,elenod_,
  elemat_,elefab_,eleopt_,u_]:= Module[{numele=Length[elenod],
  numnod=Length[nodcoor],numer,type,ne,eNL,eftab,i,ii,j,k,n,
  Em,Ν,Emat,th,ncoor,ue,ncount,esig,sige,sig},
  ncount=Table[0,{numnod}]; {numer}=eleopt;
  sige=  Table[0,{numele}]; sig=Table[{0,0,0},{numnod}];
  For [ne=1,ne<=numele,ne++,
       {type}=eletyp[[ne]]; 
       eNL=elenod[[ne]]; n=Length[eNL]; ue=Table[0,{2*n}]; 
       eftab=Flatten[Table[{2*eNL[[i]]-1,2*eNL[[i]]},{i,1,n}]];
       ncoor=Table[nodcoor[[eNL[[i]]]],{i,n}];
       Do [ii=eftab[[i]];ue[[i]]=u[[ii]],{i,1,2*n}];       
       {Em,Ν}=elemat[[ne]]; 
       Emat=Em/(1-Ν^2)*{{1,Ν,0},{Ν,1,0},{0,0,(1-Ν)/2}};
       th=elefab[[ne]]; 
       If [type=="Trig3", esig=Trig3IsoPMembraneStresses[ncoor,
          {Emat,0,0},{th},{numer},ue] ]; 
       If [type=="Quad4", esig=Quad4IsoPMembraneStresses[ncoor,
          {Emat,0,0},{th},{numer,2},ue] ];
       If [type=="Trig6", esig=Trig6IsoPMembraneStresses[ncoor,
          {Emat,0,0},{th},{numer,3},ue] ];
       If [type=="Quad9", esig=Quad9IsoPMembraneStresses[ncoor,
          {Emat,0,0},{th},{numer,3},ue] ];
       If [type=="Trig10",esig=Trig10IsoPMembraneStresses[ncoor,
          {Emat,0,0},{th},{numer,3},ue] ];
       esig=Chop[esig]; sige[[ne]]=esig; 
       For [i=1,i<=n,i++,k=eNL[[i]]; ncount[[k]]++; 
            Do[ sig[[k,j]]+=esig[[i,j]],{j,3}] ];
      ];       
  For [n=1,n<=numnod,n++,k=ncount[[n]];If [k>1,sig[[n]]/=k] 
      ];
  Return[sig]; 
];

Stress Contour Plots

Introduction to FEM

Nodal stress sig�xx Nodal stress sig�yy Nodal stress sig�xy

Stress Contour Plots  (cont'd)
Introduction to FEM

A Complete Driver Cell, Part 1: 
Model Definition Script 

Introduction to FEM

ClearAll[Em,Ν,th];  
Em=10000; Ν=.25;  th=3; aspect=6/5; Nsub=4;

(*  Define FEM model *)

ElemTypes=Table[{"Quad4"},{numele}];
NodeCoordinates=N[{{0,6},{0,0},{5,6},{5,0}}];          
ElemNodeLists=  {{1,2,4,3}};
numnod=Length[NodeCoordinates]; numele=Length[ElemNodeLists];  
ElemMaterial=   Table[{Em,Ν},{numele}]; 
ElemFabrication=Table[{th},{numele}];
FreedomValues=FreedomTags=Table[{0,0},{numnod}];
FreedomValues[[1]]=FreedomValues[[3]]={0,75}; (* nodal loads *)
FreedomTags[[1]]={1,0};     (* vroller @ node 1 *)
FreedomTags[[2]]={1,1};     (* fixed node 2 *)
FreedomTags[[4]]={0,1};     (* hroller @ node 4 *)
ElemOptions={True};

Plot2DElementsAndNodes[NodeCoordinates,ElemNodeLists,aspect,
    "One element mesh - 4-node quad",True,True];

A Complete Driver Cell,  Part 2: 
Processing Script 

Introduction to FEM

B C

DJ

75 kips 75 kips

1 3

42

�
��
�

�

�
�
�
�

1

(*  Solve problem and print results *)

{u,f,sig}=MembraneSolution[
    NodeCoordinates,ElemTypes,ElemNodeLists,
    ElemMaterial,ElemFabrication,
    ElemOptions,FreedomTags,FreedomValues];

Print["Computed displacements=",u];
Print["Recovered forces=",f];
Print["Avg nodal stresses=",sig]; 

(*  Plot Displacement Components Distribution *)

ux=uy=Table[0,{numnod}]; 
Do[ux[[n]]=u[[2*n-1]]; uy[[n]]=u[[2*n]],{n,1,numnod}];
uxmax=uymax=0;
Do[uxmax=Max[Abs[ux[[n]]],uxmax]; uymax=Max[Abs[uy[[n]]],uymax],
  {n,1,numnod}]; 
ContourPlotNodeFuncOver2DMesh[NodeCoordinates,ElemNodeLists,ux,
       uxmax,Nsub,aspect,"Displacement component ux"];
ContourPlotNodeFuncOver2DMesh[NodeCoordinates,ElemNodeLists,uy,
       uymax,Nsub,aspect,"Displacement component uy"];

(*  Plot Averaged Nodal Stresses Distribution *)

sxx=syy=sxy=Table[0,{numnod}]; 
Do[{sxx[[n]],syy[[n]],sxy[[n]]}=sig[[n]],{n,1,numnod}];
sxxmax=syymax=sxymax=0;
Do[sxxmax=Max[Abs[sxx[[n]]],sxxmax]; 
   syymax=Max[Abs[syy[[n]]],syymax];
   sxymax=Max[Abs[sxy[[n]]],sxymax],{n,1,numnod}]; 
ContourPlotNodeFuncOver2DMesh[NodeCoordinates,ElemNodeLists,
       sxx,sxxmax,Nsub,aspect,"Nodal stress sig-xx"];
ContourPlotNodeFuncOver2DMesh[NodeCoordinates,ElemNodeLists,
       syy,syymax,Nsub,aspect,"Nodal stress sig-yy"];
ContourPlotNodeFuncOver2DMesh[NodeCoordinates,ElemNodeLists,
       sxy,sxymax,Nsub,aspect,"Nodal stress sig-xy"];

A Complete Driver Cell,  Part 3: 
PostProcessing Script 

Introduction to FEM

